// File: contracts/Station3Patron.sol



// SPDX-License-Identifier: MIT

/// @title Station3 Patron  by Station3 NYC
/// @author transientlabs.xyz

/*||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||
||                                                                                        ||
||    │\││││││¡¡││¡¡││¡░¡││\│││'¡¡││¡││¡│¡'\││││││││││¡││░│░░│░┐¡│││\¡│││'¡¡││¡│░│││      ||
||    ¡¡││░░││¡│││││░│││││¡│││││││││││││¡'¡¡│││││││░││││¡│¡│¡│┐││││.│││││││░░││││¡│││░    ||
||    '││¡¡¡¡│\¡░¡│░¡¡││¡¡¡░│¡┐\│.░│░│││¡││││¡░│¡░░░││░││¡¡¡¡│││¡│¡│░││¡││││¡│││'│││¡│    ||
||    │││││¡│││¡│░│││¡│││¡│││││¡Q╓¡│¡││││'│││░│││││░│¡¡│││││¡│││¡│││││││││││''││¡│││││    ||
||    ││││░││││'¡││'│¡│¡│░││░││.╣▓▒░│¡│░¡¡¡¡┐│\│░││\││││'\│││\¡░│░¡│││░'¡¡░░¡│││¡\│¡│¡    ||
||    │¡││¡\│┐││││││¡│││││░│││¡^█▀▀¡░││││░│││¡¡░││'│││││''││░\│¡│¡││¡│││││││¡│││░░¡│││    ||
||    ││░¡│'│││░░░││'│││¡││░¡││¡▓▒│'│░┐│││││││││││░││││││││¡░│││││'││││││││││░││¡\│¡¡│    ||
||    ││││░¡¡│¡│¡¡\││││'││¡││\\╠▀▒░│││││¡││││\│││││░¡│┐│││┐│¡│¡│││░││¡\¡'│░│││¡││││┐││    ||
||    ¡│¡░│││││││░││¡¡¡│││¡│││▐▌ ,ε││││││¡││││││││││¡│││¡│││││││││││││¡░││¡│░││\\¡││││    ||
||    ││││¡¡'││┐││░││░││¡'¡│¡ε▓╙²Θ░¡¡││¡│¡¡││¡'│││'¡│¡││░¡│¡│¡││¡││¡¡││¡░¡¡'¡░│││││'││    ||
||    ¡│││¡¡¡││││││░│││¡│¡'\▐▒▒»j││││││││¡│¡\┐│││││¡│││¡¡│││'│││││¡│││¡░││¡││││░¡││││¡    ||
||    │││¡¡¡│░¡░┐¡││¡│¡││¡¡░▓█▓▒░│'''¡│░¡│░││¡│¡│░││░░\│\││'│¡'│¡│¡│¡│░\│'││░¡'¡│░│││\    ||
||    │░││¡¡│││¡¡░░│\│││││\]█╬▄▒⌐;│\¡││││¡││││││¡││││░│'¡│¡¡│░││░│¡│¡│¡│¡│││¡│││░¡¡││┐    ||
||    ¡││░¡│¡│││¡││¡'░\¡││¡▓█▓▓╠▒⌐_░¡│││││││░││¡\│¡│░││¡░│¡\░░│││¡\¡\│││¡'││\││░¡¡││¡¡    ||
||    ¡│││││││¡¡││││││'¡││j██▓▒╩╔░ ;¡│││\¡││││¡¡.\│¡┐│¡┐¡│¡│\││¡│\││░¡¡\│││░¡¡¡¡¡¡│░││    ||
||    ¡¡░¡¡░¡│¡¡¡¡│¡'││¡!░.▓█▓▓╬▒▒ ╚││¡│┐░░░¡░│¡│.│░░'¡│\│¡│¡││││░│¡¡│¡░░¡¡││¡\│¡¡│┐│¡    ||
||    │\│¡░│││¡¡│¡││░░¡\¡░\╟▓▓▓╬▒░ ╘¡\││░¡¡░¡¡¡││¡│││││░│¡││¡░░││\░││░¡░¡│││¡│\░││░¡¡░    ||
||    ¡¡││││¡░'││░│¡¡░│¡│░│▓█▓▓╬▒φ▒_]¡│││░'│¡¡¡│\│░░¡¡¡¡¡│¡│¡░│¡││¡¡\││¡│││░░\│││¡││¡¡    ||
||    ¡│░│\¡¡░¡░░¡¡│'░¡¡░░¡██▓╬╬▒╟╛╞'¡░┐░░││¡¡││¡│\'││¡││¡┐'│░\│¡░¡¡.¡\░¡│¡¡││\││░░¡░┐    ||
||    ¡\¡│¡¡¡¡¡¡░░\││\░░│░¡▓▓▓╬╠▒╢#╠_│'░\│││¡¡│¡││░░││¡¡││││░│││││\¡│¡│\│\¡││¡│¡\¡│░'░    ||
||    ░\│¡¡░░¡│¡¡\││░¡░¡│¡╓╬╙║╣╬╣▒ ▌;││¡¡│∩││!\░¡!¡░│'¡│¡░¡¡'\░│¡¡░│¡░│¡│░¡\¡┐░¡││¡¡¡│    ||
||    ¡││\¡¡'¡│'¡│░¡│¡¡\░▄▓▒╩╠╣▓╣░:▒\░│¡'¡││¡░\¡│\░\¡¡¡░│¡│¡¡¡││\│░¡││░│¡¡││░│┐│││\││¡    ||
||    ¡\¡│\░░¡│¡│░│░░░¡╓▓▓▒╠▒╗╬╟╣▒▒░░│¡¡││░¡\¡¡¡¡│░░░│¡\¡░¡││░░│¡¡░'¡¡¡¡\¡░¡░░'░¡¡\¡░¡    ||
||    ░¡\¡░░¡│τ│¡¡░¡\¡│╟▄╬▒╩▒╬╠╠▓╙╬∩¡¡░┐¡¡││░░¡││¡¡¡░¡¡¡│││¡││││░¡¡\││┐∩│││¡'││¡¡¡¡││¡    ||
||    '░¡░│¡░\╡¡¡░¡░¡'▓▓▓▓▓▓▓▓▓▓▒φ, '"│░'¡¡│░░│││░│░░¡¡││¡│\░.¡│'░┐¡¡∩││¡\│¡\┐░¡░¡¡¡¡│    ||
||    │░░¡│░│░╡░░¡│││¡█████████▓╬╩╩╠╠▒░┐░¡¡┐░¡░¡!¡¡¡\░│░¡░░░│¡¡│░░░¡│░¡¡│\¡│░░\\¡░¡¡\¡    ||
||    \¡││¡││¡▌¡│░j▓▓▓▓▓█▓██▓█▓▓▓▌#å▒░░│░│\│'¡│││\¡│¡¡│││¡¡│░││¡¡│┐░░░\¡│¡¡¡¡░¡░│░││¡¡    ||
||    ¡¡│░¡¡╓▓▓▓░\┐╙█████████████▀╬╣▒▒▒╩▒░'¡││¡\░│¡¡░░¡¡\¡░¡¡¡│¡¡│░░││░░¡¡│\░░│\░¡░¡¡│    ||
||    ░\││░¡╫▓▓╠╠¡░\╫█▓██╬██╫█╬▓▓░╠▓▐▒▌│┐¡¡││░¡¡│░│░││¡¡│░¡¡░¡│\¡¡\││¡¡░¡\││\│││¡¡│¡'│    ||
||    ░¡¡¡¡░▓██╢╬¡¡'╫█▓╫█▒█▓╫█╬▓▓░╠▓║▓▌│░░░░│!░│\│¡¡¡¡¡░¡¡'│¡¡│¡¡░░│░'¡¡│\│¡¡\░││\│¡¡¡    ||
||    ░¡¡¡¡¡▓█▓╠╠¡░░╫█▓▓█▓▓▓▓█▓▓▓▒╠░░╙╚│││││¡│¡¡│¡│░¡¡│¡│¡\¡'¡\\¡\│¡││¡¡\│¡'¡¡│││░░│¡│    ||
||    ░¡│││\▓▓╬╠╬\│\▓█▓╬╬╬╬╬╬╣▓██▒▐¡░░¡░░│░░│\░│░¡│'│¡│¡¡\░\░¡¡░¡░¡│░││░░│¡││││¡│¡│││\    ||
||    ¡░│¡┐¡▓▓╬▒╬│'▓█████████████▌╙ε░∞∞φ∩¡¡░│¡¡╟▒¡│¡¡│¡┐│││░¡░░¡│\││││\││¡¡'▒░│¡│¡¡│││    ||
||    "ⁿ""ⁿ"▓▓╬╬╠"▓███████████████▒▒▒▒"░²=""ⁿ╓▄╫▒░░││'"""""ⁿ"""""""""""""""▓╬▒""""""]▓    ||
||          ▓▓╬╬╠ ╟▓▓▓▓█▓▓▓▓▓▓▓▓▓▓░░░░░∩│╓▄▄ ╟╬█▒╟███▄,     ]▄▄▄▓╔▓⌐      j█╣▌     :█▌    ||
||    ,╓▄▄▄╦╬▓██▓▒▓▓▓▓▓█╬▓▓▒▓▓▓▓▓▓▒░▒▐▄▄░░╟▌╠╟█████████▒▓▄▄▄▄██╚▓╠▓╥µ___▌╠▒██▓,,____██    ||
||    ▓████▒╠▒▒╬╬▒╬▒▒╬╬╬▒╬▒▒╬╬╬╬╬╬▒▒░,,└,│^╙╙▀▀█▓▀╣╬╙▀█▒▓███▓▓▓▓▓╣╫▓╬▄░▒▓╟▌╠██▓▓▓▄▄╦▓█    ||
||    ▀▀╣▀▀▀╠▀╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬▀╠╠╠╩       ▀▀▀╝╝╩╠╣╫███╬╬█╬▓▒▄#▓▓╬░░▓▓▌╩▓█████▓▒▓█    ||
||    ╣▓▓▓▓▓▓▓█▓█▓▓▓▓▓▓▓█▓█▓▓▓▓▓▓▓▌▓▌▓▌▒Æ@''╓▓▀ .µ     \▐█▒╬█▄▌╬█▓▓█╩φ▓╩██▓▄▄██████▒▓█    ||
||    ▄░░░▒▄▓▓▓▓│││╠│││││││││║╬╣╣╬╬╬╣╬╬╣╣╣▄▄╣▓▄▄╣░"""│»»⌠╚╚░╙╚╙╙╙░░░░╙╙░╚╙╙╚│╚╟███▓███    ||
||    ██▒╠▓█████░░░▓▒░░░░░╟██╟███▓▓▓▓▓▓▓▓▓▓▓▓▓▓▓▓░' '░██▌φφφφφφφφφφφφ░░░░░░░░░╫███████    ||
||    ██▒▓██╬▓█╬░░░█▌░░░░╔▓██▒███▓▓▓▓▓▓▓▓▓▓▓▓▓▓▓▓░''.░██▌▒▒▒▒▒▒▒▒▒▒▒▒░░░░░░░░░╟▓▓▓▓▓▓▓    ||
||    ██▓███▌▒█▒░░╫█▌▒░░░╫███▌▒██╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬░'.┌░██▌▒▒▒▒▒▒▒▒▒▒▒▒░░░░░░░░░╔▄▄▄▄▄▄▄    ||
||    ██████▓▓▓▓▓▓▓▓╬╬╬╬▓▓▓██╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬╬    ||
||    ╚╚╚╚╩╩╩╚╚╩╚╩╚╚╩╚╚╩╚╚╩╩╩╚╚╚╚╚╚╚╚╩╩╚╩╚╩╩╚╩╚╩╚╚╚╚╚╚╚╚╚╩╚╚╩╚╚╚╚╩╚╚╚╚╚╚╩╚╩╚╚╚╚╚╚╚╩╚╚╚    ||
||    ░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░░    ||
||    ╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣╣    ||
||    '.'.'''.'''''''''''''''''''''''.'''''''''''''.''..''.'''''''''''.'''''''''''''''    ||
||    '''''''''''''''''''"''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''    ||
||                                                                                        ||
||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||*/

pragma solidity 0.8.19;

import {TLCreator} from "tl-creator-contracts/TLCreator.sol";

contract Station3Patron is TLCreator {
    constructor(
        address defaultRoyaltyRecipient,
        uint256 defaultRoyaltyPercentage,
        address[] memory admins,
        bool enableStory,
        address blockListRegistry
    )
    TLCreator(
        0x12Ab97BDe4a92e6261fca39fe2d9670E40c5dAF2,
        "Station3 Patron ",
        "STAT3",
        defaultRoyaltyRecipient,
        defaultRoyaltyPercentage,
        msg.sender,
        admins,
        enableStory,
        blockListRegistry
    )
    {}
}

// File: lib/tl-creator-contracts/src/TLCreator.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

import {ERC1967Proxy} from "openzeppelin/proxy/ERC1967/ERC1967Proxy.sol";

/*//////////////////////////////////////////////////////////////////////////
                            TLCreator
//////////////////////////////////////////////////////////////////////////*/

/// @title TLCreator.sol
/// @notice Transient Labs Core Creator Contract
/// @dev this works for either ERC721TL or ERC1155TL contracts, just need to change the implementation
/// @author transientlabs.xyz
/// @custom:version 2.3.0
contract TLCreator is ERC1967Proxy {
    /// @param name: the name of the contract
    /// @param symbol: the symbol of the contract
    /// @param defaultRoyaltyRecipient: the default address for royalty payments
    /// @param defaultRoyaltyPercentage: the default royalty percentage of basis points (out of 10,000)
    /// @param initOwner: initial owner of the contract
    /// @param admins: array of admin addresses to add to the contract
    /// @param enableStory: a bool deciding whether to add story fuctionality or not
    /// @param blockListRegistry: address of the blocklist registry to use
    constructor(
        address implementation,
        string memory name,
        string memory symbol,
        address defaultRoyaltyRecipient,
        uint256 defaultRoyaltyPercentage,
        address initOwner,
        address[] memory admins,
        bool enableStory,
        address blockListRegistry
    )
        ERC1967Proxy(
            implementation,
            abi.encodeWithSelector(
                0x1fbd2402, // selector for "initialize(string,string,address,uint256,address,address[],bool,address)"
                name,
                symbol,
                defaultRoyaltyRecipient,
                defaultRoyaltyPercentage,
                initOwner,
                admins,
                enableStory,
                blockListRegistry
            )
        )
    {}
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/proxy/ERC1967/ERC1967Proxy.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/ERC1967/ERC1967Proxy.sol)

pragma solidity ^0.8.0;

import "../Proxy.sol";
import "./ERC1967Upgrade.sol";

/**
 * @dev This contract implements an upgradeable proxy. It is upgradeable because calls are delegated to an
 * implementation address that can be changed. This address is stored in storage in the location specified by
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967], so that it doesn't conflict with the storage layout of the
 * implementation behind the proxy.
 */
contract ERC1967Proxy is Proxy, ERC1967Upgrade {
    /**
     * @dev Initializes the upgradeable proxy with an initial implementation specified by `_logic`.
     *
     * If `_data` is nonempty, it's used as data in a delegate call to `_logic`. This will typically be an encoded
     * function call, and allows initializing the storage of the proxy like a Solidity constructor.
     */
    constructor(address _logic, bytes memory _data) payable {
        _upgradeToAndCall(_logic, _data, false);
    }

    /**
     * @dev Returns the current implementation address.
     */
    function _implementation() internal view virtual override returns (address impl) {
        return ERC1967Upgrade._getImplementation();
    }
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/proxy/Proxy.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (proxy/Proxy.sol)

pragma solidity ^0.8.0;

/**
 * @dev This abstract contract provides a fallback function that delegates all calls to another contract using the EVM
 * instruction `delegatecall`. We refer to the second contract as the _implementation_ behind the proxy, and it has to
 * be specified by overriding the virtual {_implementation} function.
 *
 * Additionally, delegation to the implementation can be triggered manually through the {_fallback} function, or to a
 * different contract through the {_delegate} function.
 *
 * The success and return data of the delegated call will be returned back to the caller of the proxy.
 */
abstract contract Proxy {
    /**
     * @dev Delegates the current call to `implementation`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _delegate(address implementation) internal virtual {
        assembly {
            // Copy msg.data. We take full control of memory in this inline assembly
            // block because it will not return to Solidity code. We overwrite the
            // Solidity scratch pad at memory position 0.
            calldatacopy(0, 0, calldatasize())

            // Call the implementation.
            // out and outsize are 0 because we don't know the size yet.
            let result := delegatecall(gas(), implementation, 0, calldatasize(), 0, 0)

            // Copy the returned data.
            returndatacopy(0, 0, returndatasize())

            switch result
            // delegatecall returns 0 on error.
            case 0 {
                revert(0, returndatasize())
            }
            default {
                return(0, returndatasize())
            }
        }
    }

    /**
     * @dev This is a virtual function that should be overridden so it returns the address to which the fallback function
     * and {_fallback} should delegate.
     */
    function _implementation() internal view virtual returns (address);

    /**
     * @dev Delegates the current call to the address returned by `_implementation()`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _fallback() internal virtual {
        _beforeFallback();
        _delegate(_implementation());
    }

    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if no other
     * function in the contract matches the call data.
     */
    fallback() external payable virtual {
        _fallback();
    }

    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if call data
     * is empty.
     */
    receive() external payable virtual {
        _fallback();
    }

    /**
     * @dev Hook that is called before falling back to the implementation. Can happen as part of a manual `_fallback`
     * call, or as part of the Solidity `fallback` or `receive` functions.
     *
     * If overridden should call `super._beforeFallback()`.
     */
    function _beforeFallback() internal virtual {}
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/proxy/ERC1967/ERC1967Upgrade.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.3) (proxy/ERC1967/ERC1967Upgrade.sol)

pragma solidity ^0.8.2;

import "../beacon/IBeacon.sol";
import "../../interfaces/IERC1967.sol";
import "../../interfaces/draft-IERC1822.sol";
import "../../utils/Address.sol";
import "../../utils/StorageSlot.sol";

/**
 * @dev This abstract contract provides getters and event emitting update functions for
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967] slots.
 *
 * _Available since v4.1._
 *
 * @custom:oz-upgrades-unsafe-allow delegatecall
 */
abstract contract ERC1967Upgrade is IERC1967 {
    // This is the keccak-256 hash of "eip1967.proxy.rollback" subtracted by 1
    bytes32 private constant _ROLLBACK_SLOT = 0x4910fdfa16fed3260ed0e7147f7cc6da11a60208b5b9406d12a635614ffd9143;

    /**
     * @dev Storage slot with the address of the current implementation.
     * This is the keccak-256 hash of "eip1967.proxy.implementation" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;

    /**
     * @dev Returns the current implementation address.
     */
    function _getImplementation() internal view returns (address) {
        return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
    }

    /**
     * @dev Stores a new address in the EIP1967 implementation slot.
     */
    function _setImplementation(address newImplementation) private {
        require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
        StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
    }

    /**
     * @dev Perform implementation upgrade
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeTo(address newImplementation) internal {
        _setImplementation(newImplementation);
        emit Upgraded(newImplementation);
    }

    /**
     * @dev Perform implementation upgrade with additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCall(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        _upgradeTo(newImplementation);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(newImplementation, data);
        }
    }

    /**
     * @dev Perform implementation upgrade with security checks for UUPS proxies, and additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCallUUPS(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        // Upgrades from old implementations will perform a rollback test. This test requires the new
        // implementation to upgrade back to the old, non-ERC1822 compliant, implementation. Removing
        // this special case will break upgrade paths from old UUPS implementation to new ones.
        if (StorageSlot.getBooleanSlot(_ROLLBACK_SLOT).value) {
            _setImplementation(newImplementation);
        } else {
            try IERC1822Proxiable(newImplementation).proxiableUUID() returns (bytes32 slot) {
                require(slot == _IMPLEMENTATION_SLOT, "ERC1967Upgrade: unsupported proxiableUUID");
            } catch {
                revert("ERC1967Upgrade: new implementation is not UUPS");
            }
            _upgradeToAndCall(newImplementation, data, forceCall);
        }
    }

    /**
     * @dev Storage slot with the admin of the contract.
     * This is the keccak-256 hash of "eip1967.proxy.admin" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _ADMIN_SLOT = 0xb53127684a568b3173ae13b9f8a6016e243e63b6e8ee1178d6a717850b5d6103;

    /**
     * @dev Returns the current admin.
     */
    function _getAdmin() internal view returns (address) {
        return StorageSlot.getAddressSlot(_ADMIN_SLOT).value;
    }

    /**
     * @dev Stores a new address in the EIP1967 admin slot.
     */
    function _setAdmin(address newAdmin) private {
        require(newAdmin != address(0), "ERC1967: new admin is the zero address");
        StorageSlot.getAddressSlot(_ADMIN_SLOT).value = newAdmin;
    }

    /**
     * @dev Changes the admin of the proxy.
     *
     * Emits an {AdminChanged} event.
     */
    function _changeAdmin(address newAdmin) internal {
        emit AdminChanged(_getAdmin(), newAdmin);
        _setAdmin(newAdmin);
    }

    /**
     * @dev The storage slot of the UpgradeableBeacon contract which defines the implementation for this proxy.
     * This is bytes32(uint256(keccak256('eip1967.proxy.beacon')) - 1)) and is validated in the constructor.
     */
    bytes32 internal constant _BEACON_SLOT = 0xa3f0ad74e5423aebfd80d3ef4346578335a9a72aeaee59ff6cb3582b35133d50;

    /**
     * @dev Returns the current beacon.
     */
    function _getBeacon() internal view returns (address) {
        return StorageSlot.getAddressSlot(_BEACON_SLOT).value;
    }

    /**
     * @dev Stores a new beacon in the EIP1967 beacon slot.
     */
    function _setBeacon(address newBeacon) private {
        require(Address.isContract(newBeacon), "ERC1967: new beacon is not a contract");
        require(
            Address.isContract(IBeacon(newBeacon).implementation()),
            "ERC1967: beacon implementation is not a contract"
        );
        StorageSlot.getAddressSlot(_BEACON_SLOT).value = newBeacon;
    }

    /**
     * @dev Perform beacon upgrade with additional setup call. Note: This upgrades the address of the beacon, it does
     * not upgrade the implementation contained in the beacon (see {UpgradeableBeacon-_setImplementation} for that).
     *
     * Emits a {BeaconUpgraded} event.
     */
    function _upgradeBeaconToAndCall(
        address newBeacon,
        bytes memory data,
        bool forceCall
    ) internal {
        _setBeacon(newBeacon);
        emit BeaconUpgraded(newBeacon);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(IBeacon(newBeacon).implementation(), data);
        }
    }
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/proxy/beacon/IBeacon.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (proxy/beacon/IBeacon.sol)

pragma solidity ^0.8.0;

/**
 * @dev This is the interface that {BeaconProxy} expects of its beacon.
 */
interface IBeacon {
    /**
     * @dev Must return an address that can be used as a delegate call target.
     *
     * {BeaconProxy} will check that this address is a contract.
     */
    function implementation() external view returns (address);
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/interfaces/IERC1967.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.3) (interfaces/IERC1967.sol)

pragma solidity ^0.8.0;

/**
 * @dev ERC-1967: Proxy Storage Slots. This interface contains the events defined in the ERC.
 *
 * _Available since v4.9._
 */
interface IERC1967 {
    /**
     * @dev Emitted when the implementation is upgraded.
     */
    event Upgraded(address indexed implementation);

    /**
     * @dev Emitted when the admin account has changed.
     */
    event AdminChanged(address previousAdmin, address newAdmin);

    /**
     * @dev Emitted when the beacon is changed.
     */
    event BeaconUpgraded(address indexed beacon);
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/interfaces/draft-IERC1822.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (interfaces/draft-IERC1822.sol)

pragma solidity ^0.8.0;

/**
 * @dev ERC1822: Universal Upgradeable Proxy Standard (UUPS) documents a method for upgradeability through a simplified
 * proxy whose upgrades are fully controlled by the current implementation.
 */
interface IERC1822Proxiable {
    /**
     * @dev Returns the storage slot that the proxiable contract assumes is being used to store the implementation
     * address.
     *
     * IMPORTANT: A proxy pointing at a proxiable contract should not be considered proxiable itself, because this risks
     * bricking a proxy that upgrades to it, by delegating to itself until out of gas. Thus it is critical that this
     * function revert if invoked through a proxy.
     */
    function proxiableUUID() external view returns (bytes32);
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: lib/tl-creator-contracts/lib/openzeppelin-contracts/contracts/utils/StorageSlot.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/StorageSlot.sol)

pragma solidity ^0.8.0;

/**
 * @dev Library for reading and writing primitive types to specific storage slots.
 *
 * Storage slots are often used to avoid storage conflict when dealing with upgradeable contracts.
 * This library helps with reading and writing to such slots without the need for inline assembly.
 *
 * The functions in this library return Slot structs that contain a `value` member that can be used to read or write.
 *
 * Example usage to set ERC1967 implementation slot:
 * ```
 * contract ERC1967 {
 *     bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;
 *
 *     function _getImplementation() internal view returns (address) {
 *         return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
 *     }
 *
 *     function _setImplementation(address newImplementation) internal {
 *         require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
 *         StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
 *     }
 * }
 * ```
 *
 * _Available since v4.1 for `address`, `bool`, `bytes32`, and `uint256`._
 */
library StorageSlot {
    struct AddressSlot {
        address value;
    }

    struct BooleanSlot {
        bool value;
    }

    struct Bytes32Slot {
        bytes32 value;
    }

    struct Uint256Slot {
        uint256 value;
    }

    /**
     * @dev Returns an `AddressSlot` with member `value` located at `slot`.
     */
    function getAddressSlot(bytes32 slot) internal pure returns (AddressSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `BooleanSlot` with member `value` located at `slot`.
     */
    function getBooleanSlot(bytes32 slot) internal pure returns (BooleanSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `Bytes32Slot` with member `value` located at `slot`.
     */
    function getBytes32Slot(bytes32 slot) internal pure returns (Bytes32Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }

    /**
     * @dev Returns an `Uint256Slot` with member `value` located at `slot`.
     */
    function getUint256Slot(bytes32 slot) internal pure returns (Uint256Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
}